Brief Introduction to Matlab (mupad) symbolic mathematics toolbox.

```
////  Comments are preceded by "//"  ////////////
```

```
E^(3*x);
```

$$e^{3x}$$

```
exp(3*x);
```

$$e^{3x}$$

```
e^(3*x);
```

$$e^{3x}$$

```
float(E^1);
float(e^1);
```

2.718281828

$$e$$

```
PI;
float(PI);
float(pi);
```

$$\pi$$

3.141592654

pi

```
factor(1*x^2+3*x-10);
```

$$(x+5)\,(x-2)$$

```
eq1 := a*x^2+b*x+c=0;
solve(eq1,x);
```

$$a\,x^2+b\,x+c=0$$

$$\begin{cases} \left\{-\frac{b+\sqrt{b^2-4\,a\,c}}{2\,a},\ -\frac{b-\sqrt{b^2-4\,a\,c}}{2\,a}\right\} & \text{if } a\neq 0 \\ \left\{-\frac{c}{b}\right\} & \text{if } a=0 \wedge b\neq 0 \\ \mathbb{C} & \text{if } a=0 \wedge b=0 \wedge c=0 \\ \varnothing & \text{if } a=0 \wedge b=0 \wedge c\neq 0 \end{cases}$$

```
sol := solve(eq1,x,IgnoreSpecialCases);
```

$$\left\{-\frac{b+\sqrt{b^2-4\,a\,c}}{2\,a},\ -\frac{b-\sqrt{b^2-4\,a\,c}}{2\,a}\right\}$$

```
sol[1];
```

$$-\frac{b+\sqrt{b^2-4\,a\,c}}{2\,a}$$

```
sol[2];
```

$$-\frac{b-\sqrt{b^2-4\,a\,c}}{2\,a}$$

```
subs(eq1,x=sol[1]);
```

$$c+\frac{\left(b+\sqrt{b^2-4\,a\,c}\right)^2}{4\,a}-\frac{b\left(b+\sqrt{b^2-4\,a\,c}\right)}{2\,a}=0$$

```
simplify(%);
```

$$0=0$$

```
anames(All, User);
```

```
{eq1, sol}
```

```
eq1;sol;
```

$$a x^2 + b x + c = 0$$

$$\left\{ -\frac{b + \sqrt{b^2 - 4 a c}}{2 a}, -\frac{b - \sqrt{b^2 - 4 a c}}{2 a} \right\}$$

```
delete eq1, sol;
```

```
eq1; sol;
```

eq1

sol

```
/////  Calculus ////////////////
```

```
diff(x^2,x);
```

$$2 x$$

```
///// defining functions  ////
y := x-> 3*x^2+4*x;
y(x);
```

$$x \to 3 x^2 + 4 x$$

$$3 x^2 + 4 x$$

```
diff(y(x),x);
y'(x); //// using ' is an alternate way to take derivatives
```

$$6 x + 4$$

$$6 x + 4$$

```
int(y(x),x);  ///// indefinite integral
```

$$x^2 (x + 2)$$

```
expand(%);
```

$$x^3 + 2 x^2$$

```
int(y(x),x=a..b);  ///// definite integral
```

$$b^2 (b + 2) - a^2 (a + 2)$$

```
//////////   Plotting  ///////////////////////
```

```
plot(x^3);
```




```
plotfunc2d(sqrt(1-x^2),x=-2..2);
```




```
plotfunc2d(sqrt(1-x^2),x=-2..2,YRange=-1..1)
```




```
//// Alternate plotting commands  /////
plt1 := plot::Function2d(sqrt(1-x^2),x=-2..2,Scaling=Constrained);
```

$$\text{plot::Function2d}\left(\sqrt{-x^2+1},\ x=-2..2\right)$$

```
plot(plt1);
```




```
plt1 := plot::Function2d(sin(x),x=-2*PI..2*PI, Color=RGB::Red);
```

```
plt2 := plot::Function2d(cos(x),x=-2*PI..2*PI, Color=RGB::Blue);
plt3 := plot::Function2d(tan(x),x=-2*PI..2*PI, Color=RGB::Green);
```

plot::Function2d(sin(x), x = −2 π..2 π)

plot::Function2d(cos(x), x = −2 π..2 π)

plot::Function2d(tan(x), x = −2 π..2 π)

```
plot(plt1,plt2,plt3);
```




```
plotfunc2d(sin(x),cos(x),tan(x),x=-2*PI..2*PI);
```




```
reset();
plt1 := plot::Function3d(sin(x)*sin(y), x=-2*PI..2*PI, y=-2*PI..2*PI, Mesh=[75,75]);
```

plot::Function3d(sin(x) sin(y), x = −2 π..2 π, y = −2 π..2 π)

```
plot(plt1);
```




```
plot::Implicit3d(x^2+2*y^2+4*z^2=1,x=-1..1,y=-1..1,z=-1..1);
```

plot::Implicit3d($x^2 + 2\,y^2 + 4\,z^2 - 1,\ x = -1..1,\ y = -1..1,\ z = -1..1$)

```
plot(%);
```




```
//////////////////////////////////////////////
//////// "reset()"  resets all assignments as if you closed and reopened mupad
reset();
```